
*                       -> L_equiv
*    VIN      R0       :                OUT    @ 1000uA  27KHz/1kOhm  = 5.9mH
*     ______/\  /\  /\_:___________________    @  100uA  4.4Hz/1KOhm =  36mH
*    |        \/  \/   :                   |
*    |                  ->    IABC= 1mA->0 |        2pi*f*L=R
*   _|_           __________|\             | |\
*  /_  \        _|_     INP | \/\/\________|_| \
* // \  \       \\\   ______| /\/\/          | /| VOUT
* \   \//             |   _ |/        B_BUF1 |/ |
*  \___/           INN|    B_OTA1               |
*   _|_               |                         |
*   ///       R1      |               R2        |
*         _/\  /\  /\_|_________/\  /\  /\______|
*       _|_  \/  \/      |        \/  \/         
*       ///              |____||______ 
*                             || C1  _|_
*                                    ///                              

VIN       VIN        0     SIN(   0    10m   1000   )  AC 10m
R0        VIN        OUT   1000
R1        INN        0     100k
R2        VOUT       INN   100
C1        INN         0     1u
B_OTA1    OUT        0     I =   -1*v(VIABC)*tanh(( -v(INN) )/.052)
B_BUF1    VOUT       0     V =    v(OUT)
V_Iabc    VIABC      0     PWL    ( 0 1m 10m 0 ) 

.control
tran      1u         10m   0      1u  
set       pensize =  2
plot      v(vin)     v(out)       
op
ac        dec        10    1      1000K 
plot      db(out)    -db(vin)     title Gain2OUT@IABC1000uA
alter     V_Iabc     dc =         100u
ac        dec        10    1      1000K  
plot      db(out)    -db(vin)     title Gain2OUT@IABC100uA
.endc

.end

=====================END_OF_SPICE============================

This code works with winspice.

If a Voltage controlled Capacitor can be made, then a voltage
controlled Inductor can be made as well. 



For all types of Voltage controlled impedance, reducing
the size of IABC will be raising impedance at a fixed
frequency. That means resistors and inductors increase
in size while capacitor decrease in size. 

*                       -> L_equiv
*    VIN      R0       :                OUT    @ 1000uA  27KHz/1kOhm  = 5.9mH
*     ______/\  /\  /\_:___________________    @  100uA  4.4Hz/1KOhm =  36mH
*    |        \/  \/   :                   |
*    |                  ->    IABC= 1mA->0 |        2pi*f*L=R
*   _|_           __________|\             | |\
*  /_  \        _|_     INP | \/\/\________|_| \
* // \  \       \\\   ______| /\/\/          | /| VOUT
* \   \//             |   _ |/        B_BUF1 |/ |
*  \___/           INN|    B_OTA1               |
*   _|_               |                         |
*   ///       R1      |               R2        |
*         _/\  /\  /\_|_________/\  /\  /\______|
*       _|_  \/  \/      |        \/  \/         
*       ///              |____||______ 
*                             || C1  _|_
*                                    ///                              
                    
V_Iabc    VIABC      0     PWL    ( 0 1m 10m 0 ) 

At 1mA, the inductor is close to 6mH. 



At .1mA, the inductor is closer to 36mH.


